


HSPICE optimization example

Here's a sample circuit:



A required *biasp* value should be attained by optimization of NB2's source resistor RBP.
Here's the command file:

```
* BIAS optimization

* Netlist generated by cvpNetlist.
* manually parametrized

.include './netlist.sp'

*RRBP agnd nb2s rbias      * resistor to be optimized (exists in netlist)

.OPTIONS post

.TEMP 25C
.PARAM avdd = 5V           * opAmp supply voltage
.PARAM agnd = 0V

Vavdd avdd agnd avdd
Vagnd agnd 0 agnd
Von on agnd avdd

* Optimization statements:
.PARAM rbias = opt1(15k,.1k,20k) $ biasp generation source resistor
.dc Vavdd 0 avdd 0.5 sweep optimize=opt1 results=vbiasp model=optX
.meas dc vbiasp find v(biasp) when v(avdd)=avdd goal=3.648
.model optX opt

.END
```

The optimization needs a .dc, .tran or .ac analysis to work.

Only 4 additional lines are needed to perform the optimization of such a circuit:

1. a .PARAM declaration for the *device parameter* to be optimized (rbias), its min. and max. limit values and the step size in between, together with its *optimization name* (opt1):

```
.PARAM rbias = opt1(15k,.1k,20k) $ biasp generation source resistor
```

2. a dc *sweep* (of V_{avdd} from 0 to $avdd$) , the *optimization name* (opt1), the *results parameter name*, and the *optimization model name* declaration:

```
.dc Vavdd 0 avdd 0.5 sweep optimize=opt1 results=vbiasp model=optX
```

3. a measurement of V_{biasp} 's value when $V_{avdd} = 5V$, and a *goal* of 3.648V for this V_{biasp} :

```
.meas dc vbiasp find v(biasp) when v(avdd)=avdd goal=3.648
```

4. the *optimization call* for the model (optX):

```
.model optX opt
```

Run the analysis and obtain an optimized resistor value for RBP .
Try with different *goal* values and obtain different result values for *rbias*.